
IBM i におけるREST APIを利用したアプリ/データ連携

2

REST APIを利用することで
クライアント・アプリケーションは
自由に開発可能

Db2
for i

• ILE RPG
• ILE 

COBOL
• JAVA
• OSS

IBM iに標準装備されたWebアプリケーション・サーバーを利用して基幹データをAPI公開！

実績のある基幹業務ロジックを使用することでセキュアなデータ・アクセス環境を実現

統合Web
アプリケーション

・サーバー

Webサービ
ス・エンジン

HTTP/HTTPS

API呼び出し

IBM i に標準装備の
Webアプリケーション・サーバー

既存プログラムによ
るセキュアなデータ
・アクセス

クライアント・アプリケーションの場所も
オンプレミス/クラウドを選ばず配置可能

Db2
サーバー
ジョブ

SQL

IBM iユーザーからのFAQ 
基幹業務ロジックが、RPGⅢをメインで利用している場合
は、REST APIを利用したアプリ・データ連携ができないの
でしょうか。

下記の２つの解決策があります
1. IBM i のプログラム (RPGⅢ) からILE RPGを経由して、

REST APIサービス化する。
2. IBM i のプログラム (RPGⅢ) をILE RPGに変換して、

REST API化する。

次ページ以降に、解説します。
3

1. IBM i のプログラム (RPGⅢ) からILE RPGを経由して、
REST APIサービス化する

4

ILEプログラムで実装されたサービスをREST APIとして外部公開し、業務アプリのRPGⅢの
PGMを呼び出す(全てのRPGをILE-RPGに変更する必要はない）
RPGⅢとILE RPGは普通に相互呼び出し可能です。

HTTP
サーバー

統合アプリケーション
・サーバー（IAS）

I L E プ ロ
グ ラ ム

I L E プ ロ
グ ラ ム

ILEプロ
グ ラ ム

クライアント

REST
リクエスト

PCML

Power Systems(IBM i)

Web 
Service
エンジン Db2 

for i

RPGⅢプ
ログラム

RPGⅢプ
ログラム

RPGⅢプ
ログラム

2.  IBM i のプログラム (RPGⅢ) をILE RPGに変換して、
REST API化する

5

業務アプリのRPGⅢのソースをILE-RPGに変換して、REST API化する

RPGソース変換(CVTRPGSRC)コマンドは，RPGⅢまたはRPG/400ソース・コードをILE RPG
ソース・コードに変換することができます。

例 RPGⅢソースのRPGⅣソースへの変換（下記のようなコマンドで自動変換されます）
CVTRPGSRC FROMFILE(MYLIB/QRPGSRC) FROMMBR(XMPLE1) TOFILE(MYLIB/QRPGLESRC) 
TOMBR(*FROMMBR) EXPCPY(*NO) CVTRPT(*YES) LOGFILE(MYLIB/QRNCVTLG) 

・このコマンドはファイルMYLIB/QRPGSRC中のRPGⅢソース・メンバーXMPLE1を，
ファイルMYLIB/QRPGLESRC中の同じ名前のRPGⅣソース・メンバーに変換します。
・RPGⅢプログラムの/COPYステートメントは展開されません。これらは，RPGⅣプログラムで

/COPYステートメントとして残ります。報告書が印刷されます。
それぞれの変換の状況は，ファイルMYLIB/QRNCVTLGに配置されます。

このページ以降は、

「IBM iのREST API使用法の簡易ガイド」

になります。



IBM i のREST API使用法の
簡易ガイド

「IBM i でのREST API使用法の簡易ガイド」

目次

8

1. ILE-RPGによるREST API活用
1. [外部 → IBM i] IBM i のプログラム（ILE RPG/ILE COBOL等）を

REST APIサービス化し、外部のアプリケーションから呼び出す
2. [IBM i → 外部] IBM iからREST APIの外部アプリーケーションを呼び

出す
1. Db2 for i のHTTPメソッド機能を用いたREST APIの呼び出し
2. ILE RPGでの統合Webサービス・クライアントAPI (Transport API) の活用

2. [外部 → IBM i] DB照会をSQL記述でREST APIサービス化
3. JSONデータとDb2 for i データの変換機能



1. ILE RPGによるREST API活用

IBM i がAPIサービスと連携する2つのアプローチ

10

1. [外部 → IBM i] IBM i のILEプログラム(ILE RPG/ILE COBOL等)をREST APIサービス化し、外部
のアプリケーションから呼び出す

2. [IBM i → 外部] IBM i のILEプログラム(ILE RPG/ILE COBOL等)から、外部のAPIを呼び出す

基幹システム

REST API
呼び出し

アプリケーション

IBM Cloud Private

アプリケーション

API

API

ILEプログラム

1

2

1.  [外部 → IBM i] 
IBM i のILEプログラム(ILE RPG/ILE COBOL等)をREST API
サービス化し、外部のアプリケーションから呼び出す

11

IBM iに標準搭載されている統合アプリケーション・サーバー(IAS)で RESTをサポート
ILEプログラムで実装されたサービスをREST APIとして外部公開可能

HTTP
サーバー

統合アプリケーション
・サーバー（IAS）

ILEプログラム

ILEプログラム

ILEプログラム
クライアント

REST
リクエスト

Web 
Service
エンジン Db2 

for i

IBM i統合アプリケーション・サーバーによる
Web Serviceエンジンサポート

12

IBM i 7.2以降でサポート （IBM i OS標準機能）
IBM i 7.2 HTTP グループPTF レベル6以降

WebサービスをデプロイするIASインスタンスはWeb Administration for i
(http://<system>:2001/HTTPAdmin) からウィザードで作成可能
関連HTTPサーバー・インスタンスも自動作成

呼び出し先ILEプログラムの指定

APIアクセス・メソッド
の定義

統合Webサービス・サーバーへのRESTサービスの作成ステップ

13

4点について検討、ウィザードで指定
・RESTサービスとして展開するILEプログラムの検討
・プログラムでの処理に対するHTTPプロトコルのメソッドの検討
・URIの検討（URI : Uniform Resource Identifier）
・入出力に関する検討

入出力パラメーターとして受け入れる形式
ILEプログラムの入力パラメーターとして渡す値の指定方法
ILEプログラムの出力パラメーターとして応答コードやHTTPヘッダーを含ませるか

統合Webサービス・サーバーへのRESTサービスの作成ステップ

14

4点について検討、ウィザードで指定
・RESTサービスとして展開するILEプログラムの検討
・プログラムでの処理に対するHTTPプロトコルのメソッドの検討
・URIの検討
・入出力に関する検討

入出力パラメーターとして受け入れる形式
ILEプログラムの入力パラメーターとして渡す値の指定方法
ILEプログラムの出力パラメーターとして応答コードやHTTPヘッダーを含ませるか

• 対象プログラムでのCRUD操作の特定
• Create: データの作成
• Read: データの参照
• Update: データの更新
• Delete: データの削除

• リソースに対するCRUD操作とHTTPメソッドを対
応付ける
• Create: POST
• Read: GET
• Update: PUT
• Delete: DELETE

統合Webサービス・サーバーへのRESTサービスの作成ステップ

15

4点について検討、ウィザードで指定
・RESTサービスとして展開するILEプログラムの検討
・プログラムでの処理に対するHTTPプロトコルのメソッドの検討
・URIの検討
・入出力に関する検討

入出力パラメーターとして受け入れる形式
ILEプログラムの入力パラメーターとして渡す値の指定方法
ILEプログラムの出力パラメーターとして応答コードやHTTPヘッダーを含ませるか

URI検討のポイント
• 操作対象のリソース (データ) を名詞で表現する

• 処理内容は使用するHTTPメソッドで判別できる

• 容易に推測可能な語句を採用する



IBM i World 2021 

© IBM 2021

統合Webサービス・サーバーへのRESTサービスの作成ステップ

16

4点について検討、ウィザードで指定
・RESTサービスとして展開するILEプログラムの検討
・プログラムでの処理に対するHTTPプロトコルのメソッドの検討
・URIの検討
・入出力に関する検討

入出力パラメーターとして受け入れる形式
ILEプログラムの入力パラメーターとして渡す値の指定方法
ILEプログラムの出力パラメーターとして応答コードやHTTPヘッダーを含ませるか

• RESTサービスとしてデプロイする各プロシージャー単
位で検討が必要
• 入出力パラメーターとして受け入れる形式

• JSON (application/json)
• XML (application/xml)

• 入力パラメーターの受け渡し方式
• パス・セグメント (/cars/{color})
• マトリックス・パラメーター (/cars;color=blue)
• クエリー・パラメーター (/cars?color=blue)

• 出力パラメーターとして応答コードやHTTPヘッダーを
含めるか
• アプリとしてHTTPヘッダーを返す場合、HTTPキャッ
シュに関するヘッダー情報を返すことが多い

統合Webサービス・サーバーへのRESTサービスの作成検討例

学生情報を処理するILEサービス・プログラム STUDENTRSC
学生情報はテーブル STUDENTRSCライブラリーのSTUDENTDB に格納
提供プロシージャーは5つ

プロシージャー 処理内容

GETALL 学生情報を全件参照

GETBYID 指定した学生コードに該当の学生情報を参照

CREATE 新規の学生情報を登録

UPDATE 既存の学生情報を更新

REMOVE 指定した学生コードの学生情報を削除

STUDENTRSC/
STUDENTDB

STUDENTRSC/
STUDENTRSC

*SRVPGM

参照, 登録

更新, 削除

統合Webサービス・サーバーへのRESTサービスの作成検討例

18

検討例
・プログラムでの処理に対するHTTPプロトコルのメソッドの検討
・URIの検討
・入出力に関する検討

プロシージャー HTTPメソッド 入出力 URI

GETALL GET
• 結果はJSON形式で出力
• HTTP応答コードとヘッダーも出力に含める

/students

GETBYID GET
• 学生コードをパスセグメントで受け取る
• 結果はJSON形式で出力
• HTTP応答コードとヘッダーも出力に含める

/students/{id}

CREATE POST • HTTP応答コードとヘッダーも出力に含める /students

UPDATE PUT • HTTP応答コードを出力に含める /students

REMOVE DELETE
• 学生コードをパスセグメントで受け取る
• HTTP応答コードを出力に含める

/students/{id}

URI

/{context-root}/students

RESTサービスとして作成するILE RPGプログラム

RESTサービスプログラムのソースは固定フォーマットでも、フリー・フォーマットでもOK

統合Webサービス・サーバーへのRESTサービスの作成

20

開発環境
IBM i 7.4
・全て単一のOS環境に構成済み

HTTPサーバー
IASインスタンス
ILEサービス・プログラム
データベース・ファイル

・ILEサービス・プログラム
STUDENTRSC/STUDENTRSC
前掲検討例のSRVPGM

・HTTPサーバーは以下でリッスン
HTTP: 10073

H
T
T
P
サ
ー
バ
ー

W
S
E
R

V
IC

E
1

統合アプリケーション・
サーバー (IAS)
WSERVICE1

STUDENTRSC/
STUDENTRSC

*SRVPGM

STUDENTRSC/
STUDENTDB

*FILE

IBM i 7.4 

クライアントPC

<参考>Webサービス用IASインスタンスの作成手順

21

1. http://<system>:2001/HTTPAdmin にブラウザーでアクセス
2. 「セットアップ」タブの「新規 Web Serviceサーバーの作成」を選択

<参考>Webサービス用IASインスタンスの作成手順 (続き)

3. IASインスタンス名 (およびテキスト) を入力、「Next」ボタンをクリック
4. IASインスタンスのポートを指定、「Next」ボタンをクリック
5. IASインスタンスのサブシステムを指定、「NEXT」ボタンをクリック

<参考>Webサービス用IASインスタンスの作成手順 (続き)

6. IASインスタンスのユーザーIDを指定、「NEXT」ボタンをクリック
7. サマリーを確認して「Finish」ボタンをクリック、作成が完了

関連のHTTPサーバー・インスタンスも自動作成
HTTPポートも確認可能

<参考>RESTサービスの作成手順

1. http://<system>:2001/HTTPAdmin にブラウザーでアクセス
2. 「管理」タブ→「Application Servers」タブ→サーバーを選択
3. 「Web Services ウィザード」→「新規サービスの配置」を選択
4. 「REST」を選択し、ILE PGM Objectに書きを入力し、「Next」ボタンをクリック

<参考>RESTサービスの作成手順 (続き)

5. 「Resource name」にURI「students」を指定し、「Next」ボタンをクリック
「URI path template」にはサービス判別のための正規表現を指定可能だが、今回は「/context-

root/students」とするため指定不要

<参考>RESTサービスの作成手順 (続き)

6．ILEオブジェクト内のプロシージャーの入出力パラメーターを調整・設定後、
「Next」ボタンをクリック

宣言時にCONSTを指定したパラメーターをinputとして自動認識

<参考>RESTサービスの作成手順 (続き)

7. 各プロシージャーに対して以下を設定
・使用するHTTPメソッド
・URIパス・テンプレート (入力パラメーターをパス・セグメント等で渡す場合はここで指定)
・出力パラメーターにHTTPステータス, ヘッダーを含む場合、該当パラメーターの設定
・受け入れるメディア・タイプ (JSON, XML)
・入力パラメーターの受け渡し方法とILE側パラメーターのマッピング設定

<REMOVEプロシージャーの例>

<参考>RESTサービスの作成手順 (続き)

7.   各プロシージャーに対して以下を設定 (続き)
・使用するHTTPメソッド
・URIパス・テンプレート (入力パラメーターをパス・セグメント等で渡す場合はここで指定)
・出力パラメーターにHTTPステータス, ヘッダーを含む場合、該当パラメーターの設定
・受け入れるメディア・タイプ (JSON, XML)
・入力パラメーターの受け渡し方法とILE側パラメーターのマッピング設定

<UPDATEプロシージャーの例>

<参考>RESTサービスの作成手順 (続き)

7.  各プロシージャーに対して以下を設定 (続き)
・使用するHTTPメソッド
・URIパス・テンプレート (入力パラメーターをパス・セグメント等で渡す場合はここで指定)
・出力パラメーターにHTTPステータス, ヘッダーを含む場合、該当パラメーターの設定
・受け入れるメディア・タイプ (JSON, XML)
・入力パラメーターの受け渡し方法とILE側パラメーターのマッピング設定

<CREATE プロシージャーの例>

<参考>RESTサービスの作成手順 (続き)

7.    各プロシージャーに対して以下を設定 (続き)
・使用するHTTPメソッド
・URIパス・テンプレート (入力パラメーターをパス・セグメント等で渡す場合はここで指定)
・出力パラメーターにHTTPステータス, ヘッダーを含む場合、該当パラメーターの設定
・受け入れるメディア・タイプ (JSON, XML)
・入力パラメーターの受け渡し方法とILE側パラメーターのマッピング設定

<GETBYIDプロシージャーの例>

<参考>RESTサービスの作成手順 (続き)

7 ． 各プロシージャーに対して以下を設定 (続き)
・使用するHTTPメソッド
・URIパス・テンプレート (入力パラメーターをパス・セグメント等で渡す場合はここで指定)
・出力パラメーターにHTTPステータス, ヘッダーを含む場合、該当パラメーターの設定
・受け入れるメディア・タイプ (JSON, XML)
・入力パラメーターの受け渡し方法とILE側パラメーターのマッピング設定

<GETALLプロシージャーの例>

<参考>RESTサービスの作成手順 (続き)

8． サービスが呼び出された際に利用されるユーザー・プロファイルを指定し、「Next」
ボタンをクリック
ホスト・サーバー・ジョブQZRCSRVSの現行ユーザーとなり、実行時のオブジェクト権限等の
基点となる。
今回はデモのためIASインスタンスと同一ユーザー・プロファイルとする

<参考>RESTサービスの作成手順 (続き)

9. サービスが呼び出された際のライブラリー・リストを設定し、
「Next」ボタンをクリック

今回は特に必要ないため、追加指定はなし

<参考>RESTサービスの作成手順 (続き)

10.    受け渡すメタデータを選択し、「Next」ボタンをクリック
今回はデフォルトのまま、特に指定しない



IBM i World 2021 

© IBM 2021

<参考>RESTサービスの作成手順 (続き)

11.   サマリーを確認し、「Finish」ボタンをクリック
デプロイされると、自動的に開始される
下記は作成後の、プロパティ

<参考>IAS上のRESTサービスの呼び出し

デフォルトではコンテキスト・ルートが/web/servicesとなる
例: http://<system>:10073/web/services/students

Webブラウザーを用いて確認



IBM i World 2021 

© IBM 2021

参考情報

IBM i アプリケーション開発
IBM Redbooks: Modernizing IBM i Applications from the Database up to the User Interface and 
Everything in Between

http://www.redbooks.ibm.com/abstracts/sg248185.html?Open
REST webサービス
IBM developerWorks: Building a REST service with integrated web services server for IBM i: Part 1
https://developer.ibm.com/tutorials/i-rest-web-services-server1/

IBM developerWorks: Building a REST service with integrated web services server for IBM i: Part 2
華氏を摂氏に変換するRESTサービスのサンプル
https://developer.ibm.com/tutorials/i-rest-web-services-server2/

IBM developerWorks: Building a REST service with integrated web services server for IBM i: Part 3
学生情報をメンテナンスするRESTサービスのサンプル
https://developer.ibm.com/tutorials/i-rest-web-services-server3/

http://www.redbooks.ibm.com/abstracts/sg248185.html?Open
https://developer.ibm.com/tutorials/i-rest-web-services-server1/
https://developer.ibm.com/tutorials/i-rest-web-services-server2/
https://developer.ibm.com/tutorials/i-rest-web-services-server3/

2.  [IBM i → 外部] IBM i アプリケーション側から
REST APIサービスを呼び出す

38

IBM i アプリケーション側からREST APIサービスを呼び出す2つの方法
1. Db2 for i HTTPスカラー関数の活用
2. ILE RPGでのTransport APIの活用

IBM i (基幹システム)

REST API
呼び出し

IBM Cloud Private

アプリケーション API

SQL

ILE RPG

(1)

(2)

IBM i 上のプログラムからクラウド・サービスをコール

39

Db2 for i のHTTPメソッド機能を活用したREST API呼び出し
「SYSTOOLS」スキーマでRESTのHTTPメソッドを呼び出すスカラー関数を提供

例）ILE RPGの組み込みSQLを使って、HTTPGETBLOB関数をREST APIをGETリクエスト
し、リソースを取得

IBM i

Watson API
Language Translator

TOWATSON.SQLRPGLE
…

SYSTOOLS.HTTPGETBLOB
(:https://watson-api-

explorer.mybluemix.net/language-
translator/api/v2/translate?model_id=en-

ja&text=This%20is%20a%20test. ,’’)

CLOB

…

Db2
for i

IBM Power

HTTP GETリクエスト

リソースを取得

原文の
言語

訳文の
言語

原文

訳文

詳細はこちら：IBM i セミナー資料「ILE RPGからWatso​n APIを使用」
https://ibm.ent.box.com/s/e0zcytg2r6iw4v6op7ujpxzvryd7yht2

開発ツール
Rational Developer for i

REST API Call

https://ibm.ent.box.com/s/e0zcytg2r6iw4v6op7ujpxzvryd7yht2

Db2 for i HTTPスカラー関数

40

IBM i のデータベース機能 (Db2 for i) でHTTPメソッドを呼び出す以下のスカ
ラー関数をサポート

httpGetClob
httpGetBlob
httpPostClob
httpPostBlob
httpPutClob
httpPutBlob
httpDeleteClob
httpDeleteBlob

ライブラリーSYSTOOLSで提供しており、SQLで利用可能
アクセスしたいURLを入力として関数に引き渡し、CLOB、BLOBで値を受け取
る

Db2 for i のHTTPメソッド機能の前提条件

41

• サポートOSバージョン
IBM i 7.2以降

• 前提ライセンス・プログラム
Java™ 1.6 以降 (5761-JV1 Option 11, 12, 14, 15)

【ご参考】エラー「 SYSTOOLSのオブジェクトが見つからない」
場合の対処法

42

日本語環境の場合、下記前提PTFを適用
R720 SI64111

R730 SI64112

APAR SE66747 - OSP-DB UPDATES FOR SYSTOOLS NOT 
HAPPENING IN CCSID 290
https://www-304.ibm.com/support/docview.wss?uid=nas2SE66747

注：PTF適用後、SYSTOOLSのオブジェクトが追加されていなかった場
合、"CALL QSYS/QSQSYSIBM"を実行

https://www-304.ibm.com/support/docview.wss?uid=nas2SE66747

Watson Language Translatorを使ったILE RPGアプリケーションの
例

43

このILE RPGアプリケーションでは、Watson APIの「Language Translator」（翻訳機能）
を使用
5250画面で入力した原文を「Language Translator」で翻訳し、5250画面に結果を出力

Watson 翻訳
言語 (1= 英語 , 2= スペイン語 , 3= フランス語 , 4= イタリア語 ,  5= 日本語 )
原文の言語 : 5    訳文の言語 : 1                                                                                                                          
原文 :                                                                                                                            
バレーボール100個が注文されました。 .

.

.

. 

訳文
A total of 100 volleyball has been ordered.

SQLCode: 00000   

F3=Exit 

インプット

アウトプット

API

処理

Language Translator API

44

Language Translator APIのURL構成
必須パラメーター
text : 例）This is a test.
model_id : 例) en-ja

Request URL
https://watson-api-
explorer.mybluemix.net/language-
translator/api/v2/translate?model_id=en-
ja&text=This%20is%20a%20test.

上記URLの青太字部分をILE RPGアプリケー
ション内で3つのパラメータとして、REST呼
び出しをします。

Model_id

原文の言語ー訳文の言語を指定

text

原文を入力

Request URL

REST API呼び出しのURL

訳文の結果

Watson API を使った RPG サンプル・アプリケーション
- 表示装置ファイル

45

表示装置ファイル「TOWATSOND.DSPF」のサンプル・ソース
入力フィールド
FROMLANG：原文の言語
TOLANG：訳文の言語
FROMTEXT：原文

出力フィールド
TOTEXT：訳文
SQLCODEO ：エラー時の

SQLコード

F3キー：プログラム終了

A INDARA
A DSPSIZ(24 80 *DS3)
A CF03(03)
A R DATAR
A 1 27'Watson翻訳'
A COLOR(BLU)
A 3  1'言語 (1=英語, 2=スペイン語,'
A 3 35'3=フランス語, 4=イタリア語,'
A 3 67' 5=日本語)'
A 4  1'原文の言語:'
A FROMLANG  1 0B 4 15VALUES(1 2 3 4 5)
A EDTCDE(X)
A 4 20'訳文の言語:'
A TOLANG    1 0B 4 34VALUES(1 2 3 4 5)
A EDTCDE(X)
A 6  1'原文:'
A FROMTEXT  320O B 8  1CHECK(LC)
A 13  1'訳文'
A TOTEXT    320A O 14  1
A 19  1'SQLCode:'
A SQLCODEO  5 0O 19 10
A 24  2'F3=Exit'
A COLOR(BLU)

Watson API を使った RPG サンプル・アプリケーション
- メイン・プロシージャ -

46

ILERPG「TOWATSON.SQLRPGLE」のサンプル・ソース (1/3)

A) データ構造は言語コードの配列を
定義（コードは、画面上の原文/訳
文言語（1 =英語(en)、2 =スペイン
語(es)など）に入力された番号に対
応）

B) F3が押されるまで画面を表示しな
がらループ

C) transLate_Text()サブプロシージャを
呼び出し、原文言語コード、訳文
言語コード、原文、および、訳文
のパラメータを渡す

**free
ctl-opt option(*srcStmt: *noDebugIO) dftactGrp(*no);
dcl-F toWatsonD workstn(*ext) usage(*input: *output) indDs(WSI);
dcl-Ds WSI qualified;
F3Exit ind pos(3);

end-Ds;

dcl-Ds *n;
*n char(10) inz('enesfritja');
lang char(2) dim(5) pos(1);
end-Ds;
exec SQL
set option naming = *SQL;

exfmt dataR;
dow not WSI.F3Exit;
transLate_Text(lang(fromLang) :

lang(toLang) :
fromText :
toText);

SQLCodeO = SQLCODE;
exfmt dataR;
endDo;
*inLR = *on;

A

B

C

Watson API を使った RPG サンプル・アプリケーション
- transLate_Text()サブプロシージャ -

47

ILERPG「TOWATSON.SQLRPGLE」のサンプル・ソース (2/3)

A) RPGはCLOBデータ型を認識しないため、
textBackをSQLTypeのCLOB変数として定義

• この定義はコンパイル後2つのサブ
フィールド textBack-Len（データの
長さ）とtextBack_Data（データ）を
持つデータ構造になる

• ※「systools.httpgetclob」はデータ
タイプ「CLOB」で返す

B) 「systools.urlencode」を使用して、入力
されたテキストをエンコード

• 問題の原因となる特殊文字（＆や<
など）を同等のものに変換

dcl-Proc transLate_Text;
dcl-Pi *n;
fromLang char(2) const;
toLang char(2) const;
fromText char(320) const;
toText char(320);
end-Pi;

dcl-s str1 varchar(1000);
dcl-s str2 varchar(1000);
dcl-s textBack SQLType(CLOB: 320);
if (fromLang =toLang);

toText = fromText;
return;

endIf;

str1 =%trimR(fromText);
exec SQL

values trim(systools.urlencode(:str1, '')) into :str2;

A

B



IBM i World 2021 

© IBM 2021

Watson API を使った RPG サンプル・アプリケーション
- transLate_Text()サブプロシージャ 続き-

48

ILE RPG「TOWATSON.SQLRPGLE」のサンプル・ソース (3/3)
C) Watson APIをREST呼び出しし、翻訳を実行するURLを生成

D) 「systools.httpgetclob」を使用して、Watson APIをREST呼び出し。

• 戻り値は、先に定義したtextBackに格納される

E) 結果が返された場合は、データの長さ(textBack_Len)分のデータ(textBack_Data)を戻
り値に代入 str1 ='https://watson-api-explorer.mybluemix.net/' +

'language-translator/api/v2/translate?model_id=' +
fromLang + '-' + toLang + '&text='+ str2;

exec SQL
values char(systools.httpgetclob(:str1, ''), 256)
into :textBack;
toText = *blanks;
if (textBack_Len >0);
toText =%subSt(textBack_Data: 1: textBack_Len);
endIf;
return;
end-Proc;

C

D

E

当プログラム例では、下記リンク先の内容、サンプル・アプリケーションを使用しています
Paul Tuohy 著 「RPG TALKS TO WATSON」Copyright © 2017 IT Jungle
https://www.itjungle.com/2016/09/27/fhg092716-story01/

https://www.itjungle.com/2016/09/27/fhg092716-story01/

REST利用APIのその他の活用例
– The Weather Company

50

気象データと基幹業務データを連携し、天候が影響を与えるビジネスに活用

Weather Company Data
for IBM Cloud

現行気象状況

気象アラート

予報

REST APIsStoreI
D

Name City Country Lat Lon

1 Store_1
Montpellie
r

France 43.614934 3.908162

2 Store_2 Rochester USA 44.061604 -92.504532

3 Store_3 Osaka Japan 34.4111 135.3112

StoreID Name City Country Lat Lon Observation Feel like

1 Store_1 Montpellier France 43.614934 3.908162 快晴 11

2 Store_2 Rochester USA 44.061604 -92.504532 所により曇り 17

3 Store_3 Osaka Japan 34.4111 135.3112 小雨 21

+

JSON_TABLE

【サンプルSQL】Db2 for i テーブル上の店舗ごとの
現在の日と今後3日間の天気予報の取得

51

既存テーブルと結合しデータ
の活用が可能

ILEプログラムで組み込み
SQLを使用しREST APIを
コール

詳細はこちら：IBM i 基幹業務データと気象データの連携_2​01708.pdf

https://ibm.ent.box.com/s/cv2i7opvvzr2bu2zb3n4t1ei0sidf19p

https://ibm.ent.box.com/s/cv2i7opvvzr2bu2zb3n4t1ei0sidf19p

Weather Company Data APIsを使用したサンプルSQL 実行時

の前提条件

52

• IBM i からインターネットへの接続経路があること
• HTTPsのポートが使用可能

JSON_TABLEを使用する場合
IBM i のライセンス・プログラム、グループPTFレベルが以下を満たすこと
IBM i 7.2 Db2 PTF Group PTF SI99702 level 14以降
IBM i 7.3 Db2 PTF Group PTF SI99703 level 3 以降

参考資料
https://developer.ibm.com/articles/i-json-table-trs/

https://developer.ibm.com/articles/i-json-table-trs/

2. [IBM i → 外部] IBM i アプリケーション側から
REST APIサービスを呼び出す

53

IBM i アプリケーション側からREST APIサービスを呼び出す2つの方法
1. Db2 for i HTTPスカラー関数の活用
2. ILE RPGでのTransport APIの活用

IBM i (基幹システム)

REST API
呼び出し

IBM Cloud Private

アプリケーション API

SQL

ILE RPG

(1)

(2)

ILE RPGでの統合Webサービス・クライアントAPIの活用

54

統合Webサービス・クライアント
Apache Axisを利用したSOAP webサービスをILEプログラムから呼び出すための機能として以前より
IBM i で提供しており、新たにREST webサービスを呼び出す機能を追加
Transport API
追加されたREST webサービスの呼び出し機能

Transport API 用途

axiscTransportCreate() トランスポート・オブジェクトの生成

axiscTransportDestroy() トランスポート・オブジェクトの破棄

axiscTransportReset() トランスポート・オブジェクトを初期状態にリセット

axiscTransportSetProperty() トランスポート属性への値のセット

axiscTransportGetProperty() トランスポート属性値の取得

axiscTransportSend() トランスポートを用いたデータの送信 (バッファー). 

axiscTransportFlush() バッファー・データのトランスポートを介した送出. 

axiscTransportReceive() トランスポートからの受信. 

axiscTransportGetLastErrorCode() 直近で失敗したトランスポート処理のトランスポート・エラー・コードの取得

axiscTransportGetLastError() 直近で失敗したトランスポート処理のトランスポート・エラー・メッセージの取得

Transport APIの基本的な利用手順

55

Transport APIを利用したILE RPGからのREST webサービスの呼び出しの基本的な手順

実体はライブラリーQSYSDIRにあるサービス・プログラムQAXIS10CC
上記APIのプロトタイプ定義が/QIBM/ProdData/OS/WebServices/V1/client/include/Axis.rpgleincとして
提供されており、ILE RPGのソースに/COPYで取り込めば簡単に利用できる

1
•axiscTransportCreate() を用いてトランスポート・オブジェクトを作成

2
•axiscTransportSetProperty() を用いてHTTPメソッドやHTTPヘッダーといった属性情報をトランスポート・オブジェクトに設定

3
•axiscTransportSend() を用いて送信するデータをバッファーに入れる (POSTやPUTで送信データがある場合)

4
•axiscTransportFlush() を用いてHTTPリクエストを送信

5
•axiscTransportReceive() を用いてHTTPレスポンスを受信

6
•axiscTransportDestroy() を用いてトランスポート・オブジェクトを削除

【サンプル】Transport APIの利用手順

56

学生情報を処理するサービス・プログラムCUSTINFFRのREST webサービスを
他のIBM i 上のILE RPGプログラムTECHPGM/CUSTINFCLから呼び出し

H
T
T
P
サ

ー
バ

ー
T
E
C
H

W
S
0
1

統合Webサービ
ス・サーバー

(IWS)
TECHWS01

TECHPGM/
CUSTINFFR
*SRVPGM

TECHDB/
CUSTOMER

*FILE

IBM iIBM i

TECHPGM/
CUSTINFCL

*PGM

REST webサー
ビスの呼び出し

サンプル・プログラムの実行

57

TECHPGM/CUSTINFCLを
実行時の5250画面での表示
例

REST webサービス経由で
JSON形式のデータを取得
データがない場合はHTTPス
テータス・コード404を取得

コマンド入力
要求レベル :   4   

すべての前のコマンドおよびメッセージ :                                         

3 > CALL QCMD                                                                 
4 > CALL PGM(TECHPGM/CUSTINFCL) PARM('00010315')                              

DSPLY  == 学生情報の取得 ==                                             
DSPLY   読み取りデータ・サイズ : 198                                      
DSPLY   読み取りデータ : {"customerDs":{"custCode":"0001031               
DSPLY  5","lastName":" 西郷 ","firstName":" 隆盛 ","zipCode               
DSPLY  ":"892-0846","address":" 鹿児島県鹿児島市加治屋町 99               
DSPLY  9-999  加治屋ハイツ 999 号室 ","telephone":"099-999               
DSPLY  -9999"}}                                                           
DSPLY  HTTP ステータス・コード : 200                                      

4 > CALL PGM(TECHPGM/CUSTINFCL) PARM('99999999')                              
DSPLY  == 学生情報の取得 ==                                             
DSPLY   読み取りデータなし
DSPLY  HTTP ステータス・コード : 404                                      

終わり
コマンドを入力して，実行キーを押してください。
===> 

F3= 終了 F4=ﾌﾟﾛﾝﾌﾟﾄ F9= コマンドの複写 F10= 詳細なメッセージの除外
F11= 全画面表示 F12= 取り消し F13= 情報援助 F16=ｼｽﾃﾑ･ﾒｲﾝ･ﾒﾆｭｰ

サービス・プログラムQSYSDIR/QAXIS10CCのバインド

58

Transport APIを使用するために、提供されているプロトタイプ定義を取り込む
下記2つのどちらかの方法を選択

1. CRTRPGMODコマンドでモジュールを作成した後に、CRTPGMコマンドでQAXIS10CCをバインド
2. バインド・ディレクトリーTECHPGM/CUSTINFCLを作成、QSYSDIR/QAXIS10CCを登録し、ソースの制御仕

様書で作成したバインド・ディレクトリーを指定
CRTBNDRPGコマンドで、QSYSDIR/QAXIS10CCをバインドした状態でプログラムを作成

コマンド入力 KIWI     
要求レベル :   4   

前のコマンドおよびメッセージ : 
> CRTBNDDIR BNDDIR(TECHPGM/CUSTINFCL)                                       

バインド・ディレクトリー CUSTINFCL がライブラリー TECHPGM に作成された
。

> ADDBNDDIRE BNDDIR(TECHPGM/CUSTINFCL) OBJ((QSYSDIR/QAXIS10CC))             
ライブラリー TECHPGM のバインド・ディレクトリー CUSTINFCL に， 1 項目が
追加され， 0 項目が追加されなかった。

以降、
2の方法を

ご紹介

＜ソース＞制御仕様書でバインド・ディレクトリーの指定

59

mainプロシージャーをメイン・プロ
シージャーとして定義
制御仕様書で作成したバインド・ディ
レクトリーを指定

**free
// 全体制御 ====================================================================
ctl-opt dftactgrp(*no);
ctl-opt main(main);
ctl-opt pgminfo(*pcml:*dclcase);
ctl-opt bnddir('TECHPGM/CUSTINFCL');

// システム提供のプロトタイプ定義をコピー
/COPY /QIBM/ProdData/OS/WebServices/V1/client/include/Axis.rpgleinc

// グローバル変数
dcl-s rc int(10) inz(0);
dcl-s tranHandle pointer;

// ---------------------------------------------------------------------------
// メイン・プロシージャー
// 学生情報のREST webサービスを呼び出して取得
// ---------------------------------------------------------------------------

＜ソース＞mainプロシージャー (1/3)

60

変数 uri にREST webサービスのURIを指定
サンプル・プログラムCUSTINFCLは学生
コードを入力パラメーターとし、URIに入
力された値をつなげる

QAXIS10CCはCPPLE (ILE C++) で作られ
ているため、変数値は忘れずにヌル終端
(x’00’で終わる) にする.

指定したURIを入力パラメーターとし、API 
“axiscTransportCreate”でトランスポー
ト・ハンドルを生成

dcl-proc main;
dcl-pi *n;
p_custCode char(8) const;

end-pi;

dcl-s uri char(200);
dcl-s response char(32768);
dcl-s propBuf char(100);
dcl-s header pointer;
dcl-s bytesRead int(10) inz(0);

// トレース取得時はコメント解除
// axiscAxisStartTrace('/tmp/axistransport.log': *NULL);

// グローバル変数の初期化
clear rc;
clear tranHandle;

// 学生情報取得のURIを指定
uri = 'http://systemname:20080/web/services/customers/' + p_custCode + x'00';

// HTTPトランスポート・ハンドルの生成
tranHandle = axiscTransportCreate(uri:AXISC_PROTOCOL_HTTP11);
if (tranHandle = *NULL);
displayMessage('TransportCreate() failed');
return;

endif;

＜ソース＞mainプロシージャー (2/3)

61

属性値としてHTTPメソッドをAPI 
“axiscTransportSetProperty”を使用して指定
今回は指定した学生コードの情報照会のため、
HTTPメソッドとしてはGETを指定

GETでの情報取得で、REST webサービス側
にデータを送信しないため、今回はAPI 
“axiscTransportSend”は使用しない

属性の指定後 API “axiscTransportFlush”を用
いてHTTPリクエストを送信
リクエストの送信後は API 
“axiscTransportReceive”を使用してレスポン
スを受け取る
受け取るデータが存在する限りループして受
け取る

// HTTPメソッドを定義し、トランスポートの属性に指定
propBuf = 'GET' + X'00';
axiscTransportSetProperty(tranHandle: AXISC_PROPERTY_HTTP_METHOD: %addr(propBuf));

displayMessage('==学生情報の取得==');

// HTTPリクエストを送信
rc = axiscTransportFlush(tranHandle);
if (rc = -1);
checkError('TransportFlush()');
return;

endif;

// HTTPレスポンスからデータを受け取り、画面に表示
rc = axiscTransportReceive(tranHandle:%addr(response):%size(response):0);
if (rc = 0);
displayMessage('読み取りデータなし');

else;
dow rc > 0 AND bytesRead < %size(response);
bytesRead = bytesRead + rc;
rc = axiscTransportReceive(tranHandle:

%addr(response)+bytesRead:
%size(response)-bytesRead:
0);

enddo;
endif;

＜ソース＞mainプロシージャー (3/3)

62

今回の例では、受け取ったデータの画面
表示はdisplayMessageプロシージャーを
別途作成、利用している
データ表示が完了すると、API 
“axiscTransportDestroy”を用いて不要に
なったトランスポート・ハンドルを破棄
し、処理を終了

if (rc = -1);
checkError ('TransportReceive()');

elseif (bytesRead  > 0);
displayMessage('読み取りデータ・サイズ: ' + %char(bytesRead));
displayMessage('読み取りデータ: ' + response);

endif;

if (rc > -1);
rc = axiscTransportGetProperty(tranHandle

:AXISC_PROPERTY_HTTP_STATUS_CODE
:%addr(header));

if (rc = -1);
checkError ('TransportGetProperty()');

else;
displayMessage('HTTPステータス・コード: ' + %str(header));

endif;
endif;

// トランスポート・ハンドルの破棄
axiscTransportDestroy(tranHandle);

return;
end-proc;



IBM i World 2021 

© IBM 2021

＜ソース＞メッセージ表示プロシージャー

63

displayMessageプロシージャーを作成し、
受け取ったデータを画面に表示

// -----------------------------------------------------------
// メッセージ表示プロシージャー
// -----------------------------------------------------------
dcl-proc displayMessage;
dcl-pi *n;
p_msg varchar(5000) const;

end-pi;

dcl-c lenPartMsg const(52);

dcl-s i int(10) inz(0);
dcl-s numOfLine int(10) inz(0);
dcl-s pos int(10) inz(0);
dcl-s len int(10) inz(0);
dcl-s partMsg char(lenPartMsg) inz(*blank);

numOfLine = %len(%trim(p_msg)) / lenPartMsg + 1;

for i = 1 to numOfLine;
pos = (lenPartMsg * (i - 1)) + 1;
if (i = numOfLine);
len = %len(p_msg) - lenPartMsg * (i - 1);

else;
len = lenPartMsg;

endif;

partMsg = %subst(p_msg:pos:len);
dsply partMsg;

endfor;
end-proc;

＜ソース＞エラー・チェック・プロシージャー

64

・Transport API使用時のエラーやHTTPス
テータス・コードのハンドリングを行うプ
ロシージャーとしてcheckErrorプロシー
ジャーを使用

・トランスポート・エラー・コード、また
コードに関するメッセージを取得するAPI
としてそれぞれ
axiscTransportGetLastErrorCode、
axiscTransportGetLastErrorが用意されて
いる

・HTTPステータス・コードの取得は
axiscTransportGetPropertyで属性から読
み取る形で実装している

// ---------------------------------------------------------------------------
// エラー・チェック・プロシージャー
// ---------------------------------------------------------------------------
dcl-proc checkError;
dcl-pi *n;
p_msg varchar(5000) const;

end-pi;

dcl-s axisCode int(10);
dcl-s statusCode pointer;
dcl-s rc int(10);

// Axisエラー・コードを取得
axisCode = axiscTransportGetLastErrorCode(tranHandle);

// 画面にエラー・コードおよびメッセージを表示
displayMessage(p_msg + ' call failed: ' +

%char(axisCode) + ':' +
%str(axiscTransportGetLastError(tranHandle)));

// HTTPでのエラーの場合はHTTPステータス・コードを取得、表示
if (axisCode = EXC_TRANSPORT_HTTP_EXCEPTION);
rc = axiscTransportGetProperty(tranHandle

:AXISC_PROPERTY_HTTP_STATUS_CODE
:%addr(statusCode));

displayMessage('HTTPステータス・コード: ' + %str(statusCode));
endif;

end-proc;

Transport APIの前提条件

65

前提PTF
IBM i 7.2: SI76961, SI77246
IBM i 7.3: SI76960, SI77245

参考情報
<developerWorks: Send and receive user-defined SOAP and REST messages 
from RPG>
https://developer.ibm.com/articles/i-send-receive-user-defined-
soap-rest-messages-trs/

https://developer.ibm.com/articles/i-send-receive-user-defined-soap-rest-messages-trs/


2. [外部 → IBM i] DB照会をSQL記述で
REST APIサービス化

（IBM i 7.4 又は、 IBM i 7.3 TR6以降の機能になります）

66

REST API を使用した新しい SQL アクセス

67

RPG PGM / SRVPGM

COBOL PGM / SRVPGM

Java / C / C++

IWS 

Server

Web からIBM i へ 1つの接続を使用
– すべてHTTPベース

Db2 for i

Db2 

Server 

JobsIWSサーバーがデータベースへの接続を
処理、REST 接続で直接 SQL を使用

67

DB照会をSQL記述でRESTサービス化する手順 (1)

68

これは、Navigator for iのApplication ServerメニューにあるWeb Serviceウィザードの「新規
サービスの配置」から作成できる。

IBM i 7.4とIBM i 7.3 TR6ではウィザードステップのSpecify Web service implementation
パラメータで「WebサービスとしてのSQL」を選択することができる

DB照会をSQL記述でRESTサービス化する手順 (2)

69

では下記に示すDb2 for iのサンプルデータ（QEOLライブラリーのTOKMSPテーブルを例に、
外部からデータを参照するREST APIサービスの構成手順を追って紹介する。

DB照会をSQL記述でRESTサービス化する手順 (3)

70

サービス化の対象とするテーブルのライブラリ(QEOL)を指定

DB照会をSQL記述でRESTサービス化する手順 (4)

71

リソース名とURIパス・テンプレートを指定する

DB照会をSQL記述でRESTサービス化する手順 (5)

72

下記は2つのプロシージャを定義している例。getAllプロシージャは全レコードデータを返
すメソッドとして定義している。getByIDプロシージャはTKBANGを指定して特定のレコー
ドを返すメソッドとして定義している。

DB照会をSQL記述でRESTサービス化する手順 (6)

73

ウィザードのステップ5ではSQL処理をカスタマイズする。ここではgetByIDプロシージャのSQL result 
typeを「Single-row result set」に変更している。この記事の例では、getALLプロシージャを含め、それ
以外のパラメータはデフォルトの設定。

DB照会をSQL記述でRESTサービス化する手順(7)

74

ウィザードのステップ6では各プロシージャのメソッド情報を指定する。getAllプロシージャはデフォル
トのパラメータのままとし、getByIDプロシージャは下記のように、入力パラメータを設定してレコード
番号を特定できるように定義する。

DB照会をSQL記述でRESTサービス化する手順(8)

75

ウィザードのステップ7ではサービスの実行ユーザーIDを指定する

DB照会をSQL記述でRESTサービス化する手順 (9)

76

ウィザードのステップ8の要約画面で設定内容を確認して完了すると、サービスが起動する

DB照会をSQL記述でRESTサービス化する手順 (10)

77

生成したREST APIサービスはWebブラウザでアクセスして稼働を確認できる。Db2 for iのデータを
JSON形式で受け取ることができる。下記はGETALLメソッド.

DB照会をSQL記述でRESTサービス化する手順 (11)

78

生成したREST APIサービスはWebブラウザでアクセスして稼働を確認できる。Db2 for iのデータをJSON形式で受
け取ることができる。下記はGETBYIDメソッド。



3. JSONデータと
Db2 for iデータの変換機能

79



IBM i World 2021 

© IBM 2021

Db2 for i と JSON オブジェクトの双方向データ交換

80

Webアプリケーションで広く使われる JSON オブジェクトへの対応を強化
・Db2 for i から JSON オブジェクトをSQLで生成する機能
・JSONオブジェクトからDb2 for iへのRDBデータへ挿入する機能

SELECT JSON_OBJECT (KEY 'BANGO' VALUE TKBANG,
KEY 'CUSTNAME' VALUE TRIM(TKNAKN),
KEY 'CUSTNAMEJ' VALUE TRIM(TKNAKJ)
) AS CUSTOMER 
FROM QEOL.TOKMSP LIMIT 1;

CUSTOMER

{"BANGO":"01010","CUSTNAME":"ｱｲ ﾘﾖｶﾝ
","CUSTNAMEJ":"阿井旅館 "}

1. JSON 出力機能
JSON_OBJECT
JSON_ARRAY

2. JSON テーブル機能
JSON_TABLE

1. JSON形式への出力機能

81

JSON_OBJECT（JSONオブジェクトの生成機能）

SELECT JSON_OBJECT (KEY 'BANGO' VALUE TKBANG,
KEY 'CUSTNAME' VALUE TRIM(TKNAKN),
KEY 'CUSTNAMEJ' VALUE TRIM(TKNAKJ)
) AS CUSTOMER 
FROM QEOL.TOKMSP LIMIT 5;

（例）
QEOLライブラリー
のTOKMSPファイル

Db2 for i

JSON_ARRAY（JSON配列の生成機能）

82

SELECT JSON_ARRAY (
JSON_OBJECT(KEY 'BANGO' VALUE TKBANG),
JSON_OBJECT(KEY 'CUSTNAME' VALUE TRIM(TKNAKN)),
JSON_OBJECT(KEY 'CUSTNAMEJ' VALUE TRIM(TKNAKJ))
) AS CUSTOMER 

FROM QEOL.TOKMSP LIMIT 5;

（例）
QEOLライブラリー
のTOKMSPファイル

Db2 for i



IBM i World 2021 

© IBM 2021

2. JSONテーブル機能

83

・JSONをRDB変換処理することができるSQL関数
JSONデータをRDBの表形式に変換
変換された表をSQLで処理可能

・SELECTステートメントのFROMで使用
IFS上のJSONファイルを処理
Webサービス等で直接取得して処理

2. JSON テーブル機能
JSON_TABLEJSON

JSONテーブルの利用方法

84

基本構文

SELECT *FROM JSON_TABLE (
JSON形式データ,
‘SQL JSON パス形式’
COLUMNS (カラム名 データタイプ PATH カラム・パス

）
）AS X

・Db2 for i HTTP機能を使用
SYSTOOLS.HTTPGETCLOBでJSON形式データを取得

（例）
SYSTOOLS.HTTPGETCLOB(
’http://express.heartrails.com/api/json?method=getStations&line=JR山手線’,null)

（参考）JSONテーブルでの駅名取得 (1)

85

HeartRails Express API（駅名などをAPI でJSON形式で公開）
http://express.heartrails.com/api/json?method=getStations&line=JR山手線

http://express.heartrails.com/api/json?method=getStations&line=JR

（参考）JSONテーブルでの駅名取得（2）

86

JSON_TABLEを含むSQLを実行（ACSのSQLスクリプトで 下記を実行）
で公開URLのJSONファイルを取得

Select * from JSON_TABLE(
SYSTOOLS.HTTPGETCLOB(
‘http://express.heartrails.com/api/json?method=getStations&line=JR山手線’,null),
'$.response.station[*]'

COLUMNS(   prev VARGRAPHIC(10) PATH '$.prev',
name VARGRAPHIC(10) PATH '$.name',
next VARGRAPHIC(10) PATH '$.next'

)
) AS X;

（参考）JSONテーブルでの駅名取得（3）

87

• API公開しているJSONデータを、IBM iのデータベースに保存）

**free                                                                                                                   
ctl-opt dftactgrp(*no) main(main);                                                                                     
dcl-proc main;                                                                                                
exec sql insert into stationp
select * from JSON_TABLE(                                                                                      

SYSTOOLS.HTTPGETCLOB(                                                                                  
'http://express.heartrails.com/api/json?method=getStations&line=JR山手線

',null),          
'$.response.station[*]'                                                                                

COLUMNS(   prev VARGRAPHIC(10) PATH '$.prev',                                                      
name VARGRAPHIC(10) PATH '$.name',                                                      
next VARGRAPHIC(10) PATH '$.next'                                                       

)                                                                                           
)  AS X;                                                                                    

end-proc; 

STATIONP

Db2 for i

IBM i 統合アプリケーション・サーバーでのRESTサービス・サポー
トのメリット

88

• IBM i 同梱の標準機能なので経済的。

• RPG/COBOLメンバーもRESTサービス開発ができるので、すぐ開始。

• 既存のRPG/COBOL資産を活かしてDXができるので生産的。

• IBM i 内のアプリ・アーキテクチャーもマイクロサービス化してアジャイ

ル対応できるのでDX対応インフラへの変換が短期間。

• Db2とJSON間のやりとりはSQLで一括変換できるのでプログラミングが

簡単。

